
F07 – Linear Equations (LAPACK)

F07HGF (SPBCON/DPBCON) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F07HGF (SPBCON/DPBCON) estimates the condition number of a real symmetric positive-definite
band matrix A, where A has been factorized by F07HDF (SPBTRF/DPBTRF).

2 Specification

SUBROUTINE F07HGF(UPLO, N, KD, AB, LDAB, ANORM, RCOND, WORK,
1 IWORK, INFO)
ENTRY spbcon(UPLO, N, KD, AB, LDAB, ANORM, RCOND, WORK,
1 IWORK, INFO)
INTEGER N, KD, LDAB, IWORK(∗), INFO
real AB(LDAB,∗), ANORM, RCOND, WORK(∗)
CHARACTER∗1 UPLO

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number (in the 1-norm) of a real symmetric positive-definite band
matrix A:

κ1(A) = ‖A‖1‖A−1‖1.

Since A is symmetric, κ1(A) = κ∞(A) = ‖A‖∞‖A−1‖∞.

Because κ1(A) is infinite if A is singular, the routine actually returns an estimate of the reciprocal of
κ1(A).

The routine should be preceded by a call to F06REF to compute ‖A‖1 and a call to F07HDF
(SPBTRF/DPBTRF) to compute the Cholesky factorization of A. The routine then uses Higham’s
implementation of Hager’s method [1] to estimate ‖A−1‖1.

4 References

[1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with
applications to condition estimation ACM Trans. Math. Software 14 381–396

5 Parameters

1: UPLO — CHARACTER*1 Input

On entry: indicates whether A has been factorized as UT U or LLT as follows:

if UPLO = ’U’, then A = UT U , where U is upper triangular;
if UPLO = ’L’, then A = LLT , where L is lower triangular.

Constraint: UPLO = ’U’ or ’L’.

2: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: KD — INTEGER Input

On entry: k, the number of super-diagonals or sub-diagonals of the matrix A.

Constraint: KD ≥ 0.

4: AB(LDAB,∗) — real array Input

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the Cholesky factor of A, as returned by F07HDF (SPBTRF/DPBTRF).

5: LDAB — INTEGER Input

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HGF
(SPBCON/DPBCON) is called.

Constraint: LDAB ≥ KD + 1.

6: ANORM — real Input

On entry: the 1-norm of the originalmatrix A, which may be computed by calling F06REF. ANORM
must be computed either before calling F07HDF (SPBTRF/DPBTRF) or else from a copy of the
original matrix A.

Constraint: ANORM ≥ 0.0.

7: RCOND — real Output

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact
singularity is detected or the estimate underflows. If RCOND is less than machine precision,
then A is singular to working precision.

8: WORK(∗) — real array Workspace

Note: the dimension of the array WORK must be at least max(1,3∗N).

9: IWORK(∗) — INTEGER array Workspace

Note: the dimension of the array IWORK must be at least max(1,N).

10: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less
than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form Ax = b; the
number is usually 4 or 5 and never more than 11. Each solution involves approximately 4nk floating-point
operations (assuming n � k) but takes considerably longer than a call to F07HEF (SPBTRS/DPBTRS)
with 1 right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The complex analogue of this routine is F07HUF (CPBCON/ZPBCON).

9 Example

To estimate the condition number in the 1-norm (or infinity-norm) of the matrix A, where

A =




5.49 2.68 0.00 0.00
2.68 5.63 −2.39 0.00
0.00 −2.39 2.60 −2.22
0.00 0.00 −2.22 5.17


 .

Here A is symmetric and positive-definite, and is treated as a band matrix, which must first be factorized
by F07HDF (SPBTRF/DPBTRF). The true condition number in the 1-norm is 74.15.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F07HGF Example Program Text
* Mark 15 Release. NAG Copyright 1991.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, KDMAX, LDAB
PARAMETER (NMAX=8,KDMAX=8,LDAB=KDMAX+1)

* .. Local Scalars ..
real ANORM, RCOND
INTEGER I, INFO, J, KD, N
CHARACTER UPLO

* .. Local Arrays ..
real AB(LDAB,NMAX), WORK(3*NMAX)
INTEGER IWORK(NMAX)

* .. External Functions ..
real F06REF, X02AJF
EXTERNAL F06REF, X02AJF

* .. External Subroutines ..
EXTERNAL spbcon, spbtrf

* .. Intrinsic Functions ..
INTRINSIC MAX, MIN

* .. Executable Statements ..
WRITE (NOUT,*) ’F07HGF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, KD
IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN

*
* Read A from data file
*

READ (NIN,*) UPLO
IF (UPLO.EQ.’U’) THEN

DO 20 I = 1, N
READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))

20 CONTINUE
ELSE IF (UPLO.EQ.’L’) THEN

DO 40 I = 1, N
READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)

40 CONTINUE
END IF

*

* Compute norm of A
*

ANORM = F06REF(’1-norm’,UPLO,N,KD,AB,LDAB,WORK)
*
* Factorize A
*

CALL spbtrf(UPLO,N,KD,AB,LDAB,INFO)
*

WRITE (NOUT,*)
IF (INFO.EQ.0) THEN

*
* Estimate condition number
*

CALL spbcon(UPLO,N,KD,AB,LDAB,ANORM,RCOND,WORK,IWORK,INFO)
*

IF (RCOND.GE.X02AJF()) THEN
WRITE (NOUT,99999) ’Estimate of condition number =’,

+ 1.0e0/RCOND
ELSE

WRITE (NOUT,*) ’A is singular to working precision’
END IF

ELSE
WRITE (NOUT,*) ’A is not positive-definite’

END IF
END IF
STOP

*
99999 FORMAT (1X,A,1P,e10.2)

END

9.2 Program Data

F07HGF Example Program Data
4 1 :Values of N and KD
’L’ :Value of UPLO
5.49
2.68 5.63

-2.39 2.60
-2.22 5.17 :End of matrix A

9.3 Program Results

F07HGF Example Program Results

Estimate of condition number = 7.42E+01

